HOMEWORK #5 SOLUTION

1) NMOS: $k' \frac{W}{L_{eff}} = 16.6\ mA/V^2$

circuit ⓐ $V_{GS} = V_T + \sqrt{\dfrac{I_{DS}}{\frac{k'}{2}\frac{W}{L_{eff}}(1+\lambda V_{GS})}}$

$I_{DS} = 200\ \mu A$

By solving for V_{GS} iteratively:
(assume an initial value for V_{GS}, e.g. $V_{GS}=0$,
fill it in in the RHS of the equation,
compute the new value of V_{GS} and repeat)

$V_{GS} = 0 \rightarrow V_{GS} = 455\ mV \rightarrow V_{GS} = 449\ mV$
$\rightarrow V_{GS} = 449\ mV$

$R_{REF} = \dfrac{V_{DD} - V_{GS}}{I_{DS}} = \dfrac{1.2V - 449mV}{200\ \mu A} = 3755\ \Omega$

$r_o \approx \dfrac{1}{\lambda I_{DS}} = 25\ k\Omega$

$V_{Dsat} = 149\ mV$

↪ saturation range is $[V_{Dsat}, V_{DD}] = [149mV, 1.2V]$
(SPICE: [160 mV, 1.2V])

$\boxed{R_{out} = r_o \approx 25\ k\Omega}$

(SPICE: 27.2 kΩ)

circuit (b) $V_{SB} = R_S I = 200\,mV$

$$V_T = V_{T0} + \gamma\left(\sqrt{2\phi_f + V_{SB}} - \sqrt{2\phi_f}\right)$$
$$= 324\,mV$$

$V_{GS} - V_T = 149\,mV$ (identical to circuit (a))

$$V_{GS} = (V_{GS} - V_T) + V_T = 149\,mV + 324\,mV = 473\,mV$$

$$R_{REF} = \frac{V_{DD} - V_{GS} - \Delta V_{R_S}}{I} = \frac{1.2\,V - 473\,mV - 200\,mV}{200\,\mu A}$$

$$\boxed{R_{REF} = 2635\,\Omega}$$

$$r_o \approx \frac{1}{\lambda I_{DS}} = 25\,k\Omega \qquad g_m = \frac{2 I_{DS}}{V_{GS} - V_T} = 2.68\,mS$$

$$\chi = \gamma / 2\sqrt{2\phi_f + V_{SB}} = 0.11$$

$$R_{out} = R_S + r_o + (1+\chi)\, g_m r_o R_S$$

$\boxed{R_{out} \approx 100\,k\Omega}$ (SPICE: $103\,k\Omega$)

$$\boxed{\text{saturation range}: [\Delta V_{R_S} + V_{Dsat},\ V_{DD}] = [349\,mV,\ 1.2\,V]}$$

(SPICE: [360 mV, 1.2V])

circuit ⓒ $V_{GS_2} = 449\,mV$ (cfr. circuit ⓐ)

$$V_{SB_4} = V_{GS_2} = 449\,mV$$
$$V_{T_4} = V_{T0} + \gamma\left(\sqrt{2\phi_f + V_{SB_4}} - \sqrt{2\phi_f}\right)$$
$$= 350\,mV$$

$$V_{GS_4} - V_{T_4} = 149\,mV$$
$$V_{GS_4} = (V_{GS_4} - V_{T_4}) + V_{T_4} = 149\,mV + 350\,mV$$
$$= 449\,mV$$

$$R_{REF} = \frac{V_{DD} - V_{GS_4} - V_{GS_2}}{I} = \frac{1.2V - 499\,mV - 449\,mV}{200\,\mu A}$$

$$\boxed{R_{REF} = 1.26\,k\Omega}$$

$$r_o \approx \frac{1}{\lambda I_{DS}} = 25\,k\Omega \qquad g_m = \frac{2 I_{DS}}{V_{Dsat}} = 2.68\,mS$$

$$x_4 = \gamma / 2\sqrt{2\phi_f + V_{SB_4}} = 0.1$$

$$R_{out} = 2r_o + (1+x)\, g_m r_o^2 \approx (1+x)\, g_m r_o^2$$
$$\boxed{R_{out} \approx 1.8\,M\Omega}$$ (SPICE : $2\,M\Omega$)

$$\boxed{\text{saturation range : } [V_{GS_1} + V_{Dsat_3},\ V_{DD}] = [648\,mV,\ 1.2\,V]}$$

(SPICE : [610 mV, 1.2 V]

$$\left(V_{D_1} = V_{S_3} = V_{G_3} - V_{GS_3} = V_{G_4} - V_{GS_3} = V_{S_4} + V_{GS_4} - V_{GS_3} = V_{S_4} = V_{G_2} = V_{GS_2}\right)$$

circuit ⓓ The DC-problem is identical to that of circuit ⓒ:
$\hookrightarrow \boxed{R_{REF} = 1.26\,k\Omega}$

(Bothe problems have 2 stacked diode-connected transistors between R_{REF} and GND: M_4 and M_2 for problem 1.c, M_4 and M_1 for problem 1.d)

Because of the same reason, the output range will be the same:

$\boxed{\text{saturation range : } [V_{GS_1} + V_{Dsat_3}, V_{DD}] = [648\,mV, 1.2V]}$

(SPICE: [610m V, 1.2V]

For the output resistance, we inject a test current into the drain of M_3 and calculate all the small signal voltages caused by this current



The boxed numbers refer to the nodes, and their voltages will be called v_1, v_2, v_3, v_4

- itest will flow through M_3 into M_1 which is diode connected
  $\Rightarrow v_1 = \dfrac{i_{test}}{g_m}$

- $v_{gs_2} = v_1 \Rightarrow g_m \dfrac{i_{test}}{g_m} = i_{test}$ will flow
  <u>out</u> of node 2 into the dependent current source of M_2

- Looking down from node 2, we see r_{o_2}.
  Looking up from node 2, we see approximately $\dfrac{1}{g_m} + R_{REF} \ll r_o$

  Therefore, all of the itest current flowing out of node 2, will come from M_3 and R_{REF}.

(If you would take the bulk effect into account, you get $\dfrac{R_{REF} + 1/g_m}{1+\chi}$ looking up from node 2)

- $v_3 = -R_{REF} \cdot i_{test}$
  $v_{gs_3} = -\left(R_{REF} + \frac{1}{g_m}\right) i_{test}$



- $$V_{ds_3} = r_o \Big(\underset{\substack{\uparrow \\ \text{original itest}}}{i_{test}} + \underbrace{i_{test}\left(R_{REF} + \tfrac{1}{g_m}\right) g_m}_{-g_m v_{gs}}\Big)$$

- $\Delta V = V_{ds_3} + V_{ds_1} = \dfrac{1}{g_m} + 2r_o + g_m r_o R_{REF}$

$$\boxed{R_{out} \approx 2r_o + g_m r_o R_{REF} = 117\ k\Omega \quad (\text{SPICE: } 134\ k\Omega)}$$

```
.title homework 5 - problem 1a

.option nomod post
.include ../models.sp

m1 out in 0 0 nmos w=4u l=0.13u
m2 in in 0 0 nmos w=4u l=0.13u
rref vdd in r=3.755k
vdd vdd 0 dc=1.2
vout out 0 dc=0.5

.op
.tf i(m1) vout

.dc vout start=0 stop=1.2 step=0.01

.end
```

```
.title homework 5 - problem 1b

.option nomod post
.include ../models.sp

m1 out in s1 0 nmos w=4u l=0.13u
rs1 s1 0 r=1k
rs2 s2 0 r=1k
m2 in in s2 0 nmos w=4u l=0.13u
rref vdd in r=2.635k
vdd vdd 0 dc=1.2
vout out 0 dc=0.7

.op
.tf i(m1) vout

.dc vout start=0 stop=1.2 step=0.01

.end
```

```
.title homework 5 - problem 1c

.option nomod post
.include ../models.sp

m1 out1 in1 0 0 nmos w=4u l=0.13u
m2 in1 in1 0 0 nmos w=4u l=0.13u
m3 out in out1 0 nmos w=4u l=0.13u
m4 in in in1 0 nmos w=4u l=0.13u
rref vdd in r=1.26k
vdd vdd 0 dc=1.2
vout out 0 dc=1

.op
.tf i(m3) vout

.dc vout start=0 stop=1.2 step=0.01

.end
```

```
.title homework 5 - problem 1d

.option nomod post
.include ../models.sp

m1 out1 out1 0 0 nmos w=4u l=0.13u
m2 in1 out1 0 0 nmos w=4u l=0.13u
m3 out in out1 0 nmos w=4u l=0.13u
m4 in in in1 0 nmos w=4u l=0.13u
rref vdd in r=1.26k
vdd vdd 0 dc=1.2
vout out 0 dc=1

.op
.tf i(m3) vout

.dc vout start=0 stop=1.2 step=0.01

.end
```

```
******
 homework 5 - problem 1a


****     small-signal transfer characteristics

     i(m1)/vout                                  =   36.7195u
     input resistance at         vout            =   27.2335k
     output resistance at i(m1)                  =   27.2335k
```

```
******
 homework 5 - problem 1b


****     small-signal transfer characteristics

     i(m1)/vout                                  =    9.0106u
     input resistance at         vout            =  110.9810k
     output resistance at i(m1)                  =  110.9810k
```

```
******
 homework 5 - problem 1c


****     small-signal transfer characteristics

     i(m3)/vout                                  =  435.4626n
     input resistance at         vout            =    2.2964x
     output resistance at i(m3)                  =    2.2964x
```

```
******
 homework 5 - problem 1d


****     small-signal transfer characteristics

     i(m3)/vout                                  =    6.8154u
     input resistance at         vout            =  146.7259k
     output resistance at i(m3)                  =  146.7259k
```


homework 5 - problem 1
10x
1x
100k
10k
Result (log)
0
200m
400m
600m
800m
1
1.2
Voltage X (lin) (VOLTS)
circuit c
circuit d
circuit b
circuit a
Voltage=6.10e-01
Resistance=2.06e+06
Voltage=6.10e-01
Resistance=1.34e+05
Voltage=3.60e-01
Resistance=1.03e+05
Voltage=1.60e-01
Resistance=2.72e+04

2)



$V_{DD} = 1.2V$

$R_D = 1k\Omega$

$I_B = 400\mu A$

ⓐ Assuming M_3 is in saturation, the circuit can be simplified as follows:



$I_{DS_{1,2}} = 200\mu A$

$\hookrightarrow V_{Dsat_{1,2}} = \sqrt{\dfrac{2 I_{DS_{1,2}}}{k' W/L}} = 155\ mV$

$g_{m_{1,2}} = \dfrac{2 I_{DS}}{V_{Dsat}} = 2.58\ mS$

$r_{o_{1,2}} \approx \dfrac{1}{\lambda I_{DS_{1,2}}} = 25\ k\Omega$ $(I_{DS_{1,2}} = 200\ \mu A$

$r_{o_3} \approx 1/\lambda I_{DS_3} = 12.5\ k\Omega$ $(I_{DS_3} = 400\mu A)$

differential-mode half-circuit:



(SPICE: 1.93 kΩ)

$R_{out,dm} = 2R_D \,//\, 2r_{o_{1,2}} = 1.92\,k\Omega$

$A_{dm} = -g_{m_{1,2}}\, R_D // r_{o_{1,2}}$

$= -2.48$

(SPICE = −2.61)

(Note: this is a small-signal circuit, where the transistor stands for its small-signal model)

common-mode half-circuit:



$R_{out,cm} = R_D \,//\, \left[g_m r_{o_{1,2}} 2r_{o_3} + r_{o_{1,2}} + 2r_{o_3} \right]$

$\approx R_D = 1\,k\Omega$

(SPICE: 999.5 Ω)

$$A_{cm} = \frac{-g_{m_{1,2}} R_D}{1 + g_{m_{1,2}}\, 2r_{o_3}}$$

$\approx -\frac{R_D}{2r_{o_3}} = 0.04$

(SPICE: 0,035)

(b)

- lower range of common mode input:
  point where $V_{DS_3} = V_{Dsat_3} = \sqrt{\frac{2 I_{DS_3}}{k' W/L}} = 220\,mV$

$$V_{IC,min} = V_{Dsat_3} + V_{GS_{1,2}}$$
$$= V_{Dsat_3} + V_{T_{1,2}} + V_{Dsat_{1,2}}$$
$$= 220\,mV + 0.3\,V + 155\,mV$$
$$= 675\,mV$$

- higher range of common mode input:
  point where $V_{DS_{1,2}} = V_{Dsat_{1,2}}$
  (when M_1 and M_2 go out of saturation)

Once M_3 is in saturation, the current through each leg is approximately fixed at $200\,\mu A$.
This means that $V_{D_{1,2}}$ is fixed at

$$V_{D_{1,2}} = V_{DD} - I \cdot R_D = 1.2V - 200\,\mu A \cdot 1\,k\Omega$$
$$= 1\,V$$

M_1 and M_2 will go out of saturation when

$$V_{DS_{1,2}} \leq V_{Dsat_{1,2}} = V_{GS_{1,2}} - V_T$$
$$\Leftrightarrow V_{D_{1,2}} \leq V_{G_{1,2}} - V_T$$
$$\Leftrightarrow V_{G_{1,2}} \geq V_{D_{1,2}} + V_T = 1V + 0.3V = 1.3V$$

- $\boxed{V_{IC} \in [675\,mV,\ 1.3V]}$

(c) For $V_{ID} = 0\,V$ (and assuming M_3 is in saturation) there will be $\frac{400\mu A}{2} = 200\mu A$ flowing through each leg of the differential pair.

For $V_{ID} \neq 0$, one of the diff pair transistors M_1 & M_2 will have more current than the other one; the difference in current will get larger when $|V_{ID}|$ gets larger. This can continue until all the $400\mu A$ flows in one of the legs of the diff pair and the other leg will be shut off.

This happens for

$$|V_{ID}| = (V_{GS_{max}} - V_T) - (V_{GS_{min}} - V_T)$$

$$= \sqrt{\frac{2 \cdot 400\mu A}{k' W/L}} - 0 = 220\,mV$$

Then $V_{i,max} = 0.9V + \frac{220\,mV}{2} = 1.01\,V$

(V_D will be $1.2V - 1k\Omega \cdot 400\mu A = 1.2V - 400\,mV = 0.8V$ and $V_G < V_D + V_T = 1.1V$, so still saturation)

$V_{i,min} = 0.9 - \frac{220\,mV}{2} = 0.79\,V$

Saturation range : $\boxed{V_{ID} \in [-220\,mV, +220\,mV]}$

(d) · Vod versus Vid - plot

The theoretical shap of this curve is derived on page 220 of the textbook:

$$V_{od} = -R_D \frac{k'}{2} \frac{W}{L} V_{id} \sqrt{\frac{4 I_B}{k' W/L} - V_{id}^2}$$

· Voc versus Vic - plot

Notice how it is impossible to figure out from this plot when M_1 - M_2 go into the linear region:
V_D will remain approximately constant since there will be still 200μA flowing through each diff pair leg, regardless of whether M_1 & M_2 are in the saturation or linear region.

```
.title homework 5 - problem 2

.option nomod post
.include ../models.sp

* main circuit

m1 vo1 vi1 com com nmos w=4u l=0.13u
m2 vo2 vi2 com com nmos w=4u l=0.13u
m3 com bias 0 0 nmos w=4u l=0.13u
m4 bias bias 0 0 nmos w=4u l=0.13u
rd1 vo1 vdd r=1k
rd2 vo2 vdd r=1k

* input network

vic vic 0 dc=0.9
vid vid 0 dc=0
edi1 vi1 vic vid 0 0.5          $ +vid/2 to positive input
edi2 vic vi2 vid 0 0.5          $ -vid/2 to negative input

* output network

eco1 voc_1 voc_2 vo1 0 0.5      $ construct (vo1+vo2)/2
eco2 voc voc_1 vo2 0 0.5
voc voc_2 0 0                   $ current probe (0V v source)
fco1 vo1 gnd voc -1             $ feed current back to circuit
fco2 vo2 gnd voc -1

* supply and bias

ibias vdd bias dc=400u
vdd vdd 0 dc=1.2

* analysis statements

.op

.probe dc vod=v(vo1,vo2) voc=v(voc)
.dc vid start=-1.2 stop=+1.2 step=0.01
.dc vic start=0 stop=1.5 stop=0.01

.tf v(vo1,vo2) vid
*.tf v(voc) vic

.end
```

```
******
 homework 5 - problem 2



****     small-signal transfer characteristics
         (differential mode)

     v(vo1,vo2)/vid                              =   -2.6078
     input resistance at         vid             =  1.000e+20
     output resistance at v(vo1,vo2)             =    1.9311k


****     small-signal transfer characteristics
         (common mode)

     v(voc)/vic                                  =  -35.3444m
     input resistance at         vic             =  1.000e+20
     output resistance at v(voc)                 = 999.5331
```


Wave Symbol
D4:A0:v(vod)
homework 5 - problem 2 - differential mode output
400m
200m
0
-200m
-400m
Voltages (lin)
-500m
0
500m
Voltage X (lin) (VOLTS)
M1 shuts off
M2 shuts off


Wave Symbol
D4:A1:v(voc)
homework 5 - problem 2 - common mode output
1.2
1.18
1.16
1.14
1.12
1.1
1.08
1.06
1.04
1.02
1
980m
Voltages (lin)
0
500m
1
1.5
Voltage X (lin) (VOLTS)
M3 cut off
M3 in linear region
M3 in saturation